@echo off
chcp 65001

echo 当前目录为：
cd
echo 当前目录下文件列表：
dir

:: 检查 WinSW 原始文件是否存在
if exist WinSW-x64.exe (
    echo WinSW 安装包已存在，跳过下载...
) else (
    echo 下载 WinSW...
    powershell -Command "Invoke-WebRequest -Uri 'https://github.com/winsw/winsw/releases/download/v3.0.0-alpha.11/WinSW-x64.exe' -OutFile 'WinSW-x64.exe'"
)

:: 如果服务程序不存在，则复制一份
if exist xxl-job-service.exe (
    echo 服务程序已存在...
) else (
    echo 创建服务程序...
    copy WinSW-x64.exe xxl-job-service.exe
)

echo 创建日志目录...
if not exist logs mkdir logs

echo 检查配置文件...
if not exist xxl-job-service.xml (
    echo 错误：找不到 xxl-job-service.xml 配置文件！
    echo 请确保 xxl-job-service.xml 文件存在于当前目录。
    pause
    exit /b 1
)

:: 检查 JAR 包路径
set JAR_PATH=..\\xxl-job-admin\\target\\xxl-job-admin-3.1.1-SNAPSHOT.jar
if exist %JAR_PATH% (
    echo 检测到 JAR 包：%JAR_PATH%
) else (
    echo 错误：未找到 JAR 包 %JAR_PATH%
    pause
    exit /b 1
)

echo 安装服务...
xxl-job-service.exe install
if %errorlevel% neq 0 (
    echo 服务安装失败，错误码：%errorlevel%
    pause
    exit /b 1
)

echo 启动服务...
xxl-job-service.exe start
if %errorlevel% neq 0 (
    echo 服务启动失败，错误码：%errorlevel%
    pause
    exit /b 1
)

echo 完成！
pause 